
$ontext 
Title:   PJM_10node_conjecture.gms 
References: Y. Chen and B.F. Hobbs, "An Oligopolistic Power Market with 
            NOx Tradable Permits," IEEE Transactions on Power System, 
            20(1): 119-120, 2005 
Author:  Yihsu Chen, The Johns Hopkins University 
Content: The code is writen for PJM electricity market model. In this model, 
         some generators are assumed to exercise Cournot strategy in elec. 
         market, while also allowing to exercise conjectured price response in 
         NOx permit market. They are represented by st(f), and rpc(f), respec- 
         tively, in the model.  The NOx cap is represented as a constraint, and 
         SO2, acid rain title 4, is represented by multipling with SO2 permit 
         price in supply curve. 
Output:  ps.txt: producer surplus, nox trade, firm total sale, gen. cost, tot 
         nox in OTC cap, tot NOx and firm capacity share 
         ts_lcp.txt:  total sale of firm f at i node 
         x_lcp.txt: individual generator output level 
         pe_lcp.txt: price level of i at t period 
         flow_lcp.txt: flow in the arcs 
         w_lcp.txt: wheeling charge 
         y_lcp.txt: outflow from node i 
         pn_lcp.txt: permit price, tot OTC NOx and tot.NOx 
Remark:  While rpc(f)=0 for all firms, the model reduces to "no conjectur price" 
         While, in addition, st(f)=0 for all firms, the model reduces to perfect 
         competition. 
Paper:   "An Oligopolistic Energy Market Model with Tradable NOx permits", by 
         Yihsu Chen and Benjamin F. Hobbs 
$offtext 
$inlinecom/* */ 
set 
h plant index /1*879/ 
i node /ME1,ME2,PPL1,BGEPEP,PPL2,DPL,PE,PS,AE,PPL3,JC1,JC2,BGE2,PN/ 
f firm /1*9/ 
t time period /1*5/ 
k lines /1*18/ 
ll /up,lo/ 
p(f,i,h) plant dynamic set 
a attributs /SCAP,COST,NOX,SO2,OTC,PCAID,OWNERID,ARAIN,MAXOfRes/; 
$include c:\My Stuffs\My 
Research\IEEE\Model\Revised\Data_10node_network_conj_1208_revised.txt 
*$include g:\ieee\Data_10node_network_conj_1208.txt 
qnox(f)=qnox(f)*0.8; /* 80% of permit used for compliance */ 
table 
d(h,a) plant h 
$include c:\My Stuffs\My 
Research\IEEE\Model\Revised\Data_10node_plant_res_revised.txt 



*$include g:\ieee\Data_10node_plant_res.txt 
; 
*Define dynamic set* 
p(f,i,h)=yes$(d(h,"PCAID")=ord(i) and d(h,"OWNERID")=ord(f)); 
 
parameter 
cap(i) total cap in node 
fcap(f) firm capacity 
e(f,i,h) emission rate 
c(f,i,h) variable cost 
st(f) strategy index /1 0,2 0,3 0,4 0,5 0,6 0,7 0,8 0,9 0/ 
npc(f) conjecture price slpe /1 0,2 0,3 0,4 0,5 0,6 0,7 0,8 0,9 0/ 
fc(f,i,t) forward contract amount 
rm reserve margin /0.075/; 
**calculation** 
e(f,i,h)$p(f,i,h)=d(h,"nox")/2000; 
cap(i)=sum(p(f,i,h),d(h,"SCAP")); 
c(f,i,h)$p(f,i,h)=d(h,"cost")+140*d(h,"SO2")*d(h,"arain"); 
fcap(f)=sum(h$(d(h,"OWNERID")=ord(f)),d(h,"SCAP")); 
 
fc(f,i,t) = ffc(i,f)*load(i,t); 
*fc(f,i,t) =0; 
 
 
 
variables 
y(i,t)  flow from hub to n at t period 
ts(i,t) total s in n at period t 
w(i,t)  wheeling charge from hub to n at t period 
theta(f,t) dual of f firm's s at t period 
pe(i,t) price of power 
fnox(f) firm nox emission 
flow(k,t) flow 
totn(f) total nox 
aa(i,t) arbitrage 
ph(t) hub price 
sps(i,t) spot sales; 
 
 
*aa.fx(i,t)=0; 
positive variables 
x(f,i,h,t) output of p plant in t period 
s(f,i,t) s of firm f at node n in t period 
rho(f,i,h,t) dual of plant h cap 
lambda_m(k,t) dual of ptdf lower 
lambda_p(k,t) dual of ptdf upper 



pn price of permits 
r(f,i,h,t) reserve capacity 
eta(f,i,h,t) dual of reserve 
pr(t) price of reserve; 
 
equations 
f1(f,i,t) cp for s 
f2(f,i,h,t) cp for x 
f3(f,i,h,t) cp for rho 
f4(f,t) cp for theta 
g1(i,t) cP for y 
g2(k,t) cp for ptdf negative lambda_n 
g3(k,t) cp for ptdf positive lambda_p 
mc1(i,t) market clearing for power 
mc2 cp for pn 
df1(i,t) price of electricity 
df2(f) firm total NOx emission 
df3(i,t) total sales 
df4(k,t) flow 
g4(f,i,h,t) cp for r 
g5(f,i,h,t) cp for eta 
mc4(t) market clearning for reserve 
a1(i,t) arbitrage 
a2(t) sum equal to zero 
df5(i,t) spot sales; 
 
f1(f,i,t).. -pe(i,t)+w(i,t)+(p0(i,t)/q0(i,t))*(s(f,i,t)-fc(f,i,t))$st(f)+theta(f,t)=g=0; 
f2(f,i,h,t)$p(f,i,h).. c(f,i,h)-w(i,t)+[pn+npc(f)*(fnox(f)-
qnox(f))]*e(f,i,h)$(d(h,"otc")=1)+rho(f,i,h,t)/b(t)-theta(f,t)=g=0; 
f3(f,i,h,t)$p(f,i,h).. -x(f,i,h,t)-r(f,i,h,t)+d(h,"scap")=g=0; 
f4(f,t).. sum(p(f,i,h),x(f,i,h,t))-sum(i,s(f,i,t))=e=0; 
g1(i,t).. b(t)*w(i,t)+sum(k,ptdf(k,i)*(lambda_p(k,t)-lambda_m(k,t)))=e=0; 
g2(k,t).. flow(k,t)+lcap(k,"lo")=g=0; 
g3(k,t).. -flow(k,t)+lcap(k,"up")=g=0; 
mc1(i,t).. y(i,t)=e=sum(p(f,i,h),x(f,i,h,t))+z(i,t)-sum(f,s(f,i,t))-aa(i,t); 
mc2..   sum(f,qnox(f))-sum(f,fnox(f))=g=0; 
df1(i,t).. pe(i,t)=e=p0(i,t)-(p0(i,t)/q0(i,t))*(ts(i,t)+aa(i,t)); 
df2(f).. fnox(f)=e=sum((p(f,i,h),t)$(d(h,"otc")=1),x(f,i,h,t)*b(t)*e(f,i,h)); 
df3(i,t).. ts(i,t)=e=sum(f,s(f,i,t)); 
df4(k,t).. flow(k,t)=e=sum(i,ptdf(k,i)*y(i,t)); 
g4(f,i,h,t)$p(f,i,h)..-pr(t)*b(t)+rho(f,i,h,t)+eta(f,i,h,t)=g=0; 
g5(f,i,h,t)$p(f,i,h).. -r(f,i,h,t)+d(h,"scap")*d(h,"MAXOfRes")=g=0; 
mc4(t)..sum(p(f,i,h),r(f,i,h,t))- sum(i,ts(i,t))*rm =g=0; 
a1(i,t).. pe(i,t)-w(i,t)=e=ph(t); 
a2(t).. sum(i,aa(i,t))=e=0; 
df5(i,t).. sps(i,t)=e=ts(i,t)-sum(f,fc(f,i,t)); 



 
 
model conj 
/f1.s,f2.x,f3.rho,f4,g1,g2.lambda_m,g3.lambda_p,mc1,mc2.pn,df1,df2,df3,df4,g4.r,g5.eta
,mc4.pr,a1,a2,df5/; 
conj.optfile=1; 
solve conj using MCP 


